


【B-1(2)】IchigoJam で信号機を作ろう (Ver.2)



●今回の目標

LED 信号機を IchigoJam につないで、
BASIC プログラムで光らせます。





●信号機の取り付けと確認

IchigoJam のソケットに、信号機の端子をさします。
信号機の端子に、GND、Out1、Out2、Out3 と書かれているので、
IchigoJam のそれぞれの端子にさします。



回路図はこのようになります。



信号機が取り付けられたら、ちゃんと LED が光るか、ダイレクトモードで確認しましょう。

```
OUT 1,1  
OUT 2,1  
OUT 3,1
```

信号機の青・黄・赤の LED が光ります。
もし光らなかったら、信号機の
さし方をまちがえています。
よく見直しましょう。




```
OUT 1,0  
OUT 2,0  
OUT 3,0
```

3 個の LED が消えます。



いろいろなパターンで信号機を光らせてみましょう。

では、信号機を光らせるプログラムを作ります。
信号機の動きを考えてみます。



今回は、青・黄・赤の点灯時間を上のようになしてみよう。

10	^*TRAFFIC LIGHT	プログラムのタイトル
20	OUT 0	LED を全部消す
30	^*LOOP	ループの始め
40	OUT 1,1	青 LED を光らせる
50	WAIT 600	10 秒待つ
60	OUT 1,0	青 LED を消す
70	OUT 2,1	黄 LED を光らせる
80	WAIT 120	2 秒待つ
90	OUT 2,0	黄 LED を消す
100	OUT 3,1	赤 LED を光らせる
110	WAIT 600	10 秒待つ
300	OUT 3,0	赤 LED を消す
310	GOTO 30	30 行へもどる

「**RUN**」でプログラムを実行してみよう。
信号機が、青→黄→赤→青→…と動きます。



WAIT (ウェイト) 命令の数値を変えると、それぞれの LED の点灯時間が変わります。
試してみよう。

「**SAVE 0**」でプログラムを保存しておきましょう。


● 歩行者用信号機を光らせる

歩行者用信号機を IchigoJam に追加します。
IchigoJam のソケットに、信号機のリード線をさします。
以下のようにさしてください。

IchigoJam の端子	歩行者用信号機の リード線
GND	黒
OUT5	赤
OUT6	青



回路図は下のようになります。



まずは歩行者用信号機の LED がちゃんと光るか、ダイレクトモードで確認しましょう。

```
OUT 5,1
OUT 6,1
```

歩行者用信号機の赤・青の LED が光ります。
もし光らない LED があったら、
どこか差し方をまちがえています。
よく見直しましょう。

```
OUT 5,0
OUT 6,0
```

2 個の LED が消えます。



いろいろなパターンで、車用信号機と歩行者用信号機を光らせてみましょう。

では、車用信号機と歩行者用信号機を動かすプログラムを作ります。
2 つの信号機の動きをどうすればいいか、考えてみましょう。

車用信号機							
歩行者用信号機							
時間	10秒	2秒	2秒	5秒	0.5秒	0.5秒	2秒

2 つの信号機の動きと時間を、表にまとめてみました。動きを説明します。

- 車用信号機を「青→黄→赤」と動かします。この間、歩行者用信号機はずっと「赤」です。
- 車用信号機を「赤」にして、2 秒待ってから、歩行者用信号機を「青」にします。
(すぐに「青」にしてしまうと、まだ車が来るかもしれないので危ないです)
- 歩行者用信号機の「青」を 5 秒間続けます。
- 歩行者に注意をうながすため、歩行者用信号機の「青」を 3 回点滅させます。
- 歩行者用信号機を「赤」にして、2 秒待ってから、最初にもどって車用信号機を「青」にします。
(すぐに「青」にしてしまうと、まだ歩行者が歩いているかもしれないので危ないです)


それでは、前ページの表と説明を見ながら、プログラムを改造してみましょう。

10	/*TRAFFIC LIGHT	
20	OUT 0	
30	/*LOOP	
40	OUT 1,1	
45	OUT 5,1	歩行者用信号の赤を光らせる
50	WAIT 600	
60	OUT 1,0	
70	OUT 2,1	
80	WAIT 120	
90	OUT 2,0	
100	OUT 3,1	
110	WAIT 120	車用信号の赤の時間を2秒に変更
120	OUT 5,0	歩行者用信号の赤を消して青を光らせる
130	OUT 6,1	
140	WAIT 300	5秒待つ
150	FOR I=1 TO 3	3回くりかえし
160	OUT 6,1	歩行者用信号の青を光らせる
170	WAIT 30	0.5秒待つ
180	OUT 6,0	歩行者用信号の青を消す
190	WAIT 30	0.5秒待つ
200	NEXT	もどってくりかえし
210	OUT 5,1	歩行者用信号の赤を光らせる
220	WAIT 120	2秒待つ
300	OUT 3,0	
310	GOTO 30	

プログラムを実行してみましょう。
2つの信号機が連動して動きます。


WAIT (ウェイト) 命令の数値を変えると、それぞれの LED の点灯時間が変わります。
試してみましょう。

「SAVE 0」でプログラムを保存しておきましょう。



●押しボタン信号にする

プログラムを改造して、「ボタンを押したら歩行者用信号が青になる」信号にしてみましょう。
信号機の動きを図で考えます。



車用信号機が青の時に、ボタンが押されているかをチェックして、押されるまではそのまま変化させません。

ボタンが押されたら、黄色以降の処理に移ります。それ以降の動きは同じです。

ボタンは、IchigoJam の左側に
付いているボタンを使います。



プログラムを改造します。

車用信号機を青にした後の、時間待ち(WAIT)の行を書きかえます。

```
10  ' *TRAFFIC LIGHT
```

```
20  OUT 0
```

```
30  ' *LOOP
```

```
40  OUT 1,1
```

```
45  OUT 5,1
```

ボタンが押されていないかったら、この行をくりかえし

```
50  IF BTN()=0 THEN GOTO 50
```

```
55  WAIT 180
```

3 秒待つ

```
60  OUT 1,0
```

```
70  OUT 2,1
```

```
80  WAIT 120
```

```
90  OUT 2,0
```

```
100 OUT 3,1
```

```
110 WAIT 120
```

```
120 OUT 5,0
```

```
130 OUT 6,1
```

```
140 WAIT 300
```

```
150 FOR I=1 TO 3
```

```
160 OUT 6,1
```

```
170 WAIT 30
```

```
180 OUT 6,0
```

```
190 WAIT 30
```

```
200 NEXT
```

```
210 OUT 5,1
```

```
220 WAIT 120
```

```
300 OUT 3,0
```

```
310 GOTO 30
```

プログラムを実行してみましょう。

車用信号がずっと青のままになります。

ボタンを押すと、3 秒後に黄色→赤に変わります。

※車用信号が青になった直後にボタンが押された場合、すぐに黄色になっても車は急に止まれないので、3 秒待つようにしています。

●「おまちください」とお知らせする

ボタンを押してから3秒待たされると、「ちゃんとボタンが押せたのか？」と不安になります。現実の押しボタン信号では、「おまちください」のランプが点いて歩行者にお知らせします。IchigoJam についている LED を、お知らせランプとして点灯させましょう。

```

10  ' *TRAFFIC LIGHT
20  OUT 0
30  ' *LOOP
40  OUT 1,1
45  OUT 5,1
50  IF BTN(<)=0 THEN GOTO 50
52  LED 1      LEDを光らせる
55  WAIT 180
60  OUT 1,0
70  OUT 2,1
80  WAIT 120
90  OUT 2,0
100 OUT 3,1
110 WAIT 120
120 OUT 5,0
130 OUT 6,1
135 LED 0      LEDを消す
140 WAIT 300
150 FOR I=1 TO 3
160 OUT 6,1
170 WAIT 30
180 OUT 6,0
190 WAIT 30
200 NEXT
210 OUT 5,1
220 WAIT 120
300 OUT 3,0
310 GOTO 30

```

プログラムを実行してみましょう。
 ボタンを押すと、IchigoJam の LED が光ってお知らせします。
 歩行者用信号が青になると、LED が消えます。

「SAVE 0」でプログラムを保存しておきましょう。